In[1]:=
```
SetDirectory["~/writing/WIP/KappaLib/"];
<< kappaLib.m
```

```
KappaLib v1.1
```

Proof of Proposition 2.2

Extra routines

In[3]:=
```
(*
 *  Convert a list of expressions into equations.
 *)
toEqs[c_] := Table[c[[i]] == 0, {i, 1, Length[c]}]
```

In[4]:=
```
(* return conditions on kappa that must be
 satisified if kappa is of purely skewon type (kappa in W) *)
emSkewonConditions[kappa_] := Module[
  {conditions, i, j, l, m, p, q},

  conditions = Table[
    Sum[emReadNormal[kappa, i, j, l, m] Signature[{l, m, p, q}], {l, 1, 4}, {m, 1, 4}]
     +
     Sum[emReadNormal[kappa, p, q, l, m] Signature[{l, m, i, j}], {l, 1, 4}, {m, 1, 4}],
    {i, 1, 4}, {j, 1, 4}, {p, 1, 4}, {q, 1, 4}
   ];

  Union[Flatten[conditions]]

 ]

(* return conditions on kappa that must be satisified
 if kappa is of purely principal type (kappa in Z) *)
emPrincipalConditions[kappa_] := Module[
  {conditions, i, j, l, m, p, q},

  conditions = Table[
    Sum[emReadNormal[kappa, i, j, l, m] Signature[{l, m, p, q}], {l, 1, 4}, {m, 1, 4}]
     -
     Sum[emReadNormal[kappa, p, q, l, m] Signature[{l, m, i, j}], {l, 1, 4}, {m, 1, 4}],
    {i, 1, 4}, {j, 1, 4}, {p, 1, 4}, {q, 1, 4}
   ];

  Append[Union[Flatten[conditions]], emTrace[kappa]]

 ]
```

Claim 1: σ(W') subset W

Step 1: Let eta0 be trace-free (1,1)-tensor

In[6]:=
$$\texttt{eta} = \begin{pmatrix} w11 & w12 & w13 & w14 \\ w21 & w22 & w23 & w24 \\ w31 & w32 & w33 & w34 \\ w41 & w42 & w43 & w44 \end{pmatrix};$$

```
eta0 = eta - 1 / 4 Tr[eta] IdentityMatrix[4];
eta0 // MatrixForm
```

Out[8]//MatrixForm=

$$\begin{pmatrix} w11 + \frac{1}{4}\,(-w11 - w22 - w33 - w44) & w12 & w13 \\ w21 & w22 + \frac{1}{4}\,(-w11 - w22 - w33 - w44) & w23 \\ w31 & w32 & w33 + \frac{1}{4}\,(-w11 - w22 - w33 - w44) \\ w41 & w42 & w43 \end{pmatrix}$$

In[9]:=
```
Simplify[Tr[eta0]]
```
Out[9]= 0

- **Step 2: Define $\sigma(\eta)$**

In[10]:=
```
sigma[w_] := Module[

   {res, i, j, l, m},

   res = emZeroKappa[];

   For[i = 1, i ≤ 4, i++,
    For[j = i + 1, j ≤ 4, j++,
      For[l = 1, l ≤ 4, l++,
        For[m = l + 1, m ≤ 4, m++,

          res[[i]][[j]][[l]][[m]] = 1 / 2 (
             w[[i]][[l]] KroneckerDelta[j, m]
              - w[[j]][[l]] KroneckerDelta[i, m]
              - w[[i]][[m]] KroneckerDelta[j, l]
              + w[[j]][[m]] KroneckerDelta[i, l]
            )

         ];
       ];
     ];
   ];

   res

  ]
```

- **Step 3: Show that $\sigma(\eta)$ is purely skewon.**

In[11]:=
```
kappa = sigma[eta0];
```
In[12]:=
```
emSkewonConditions[kappa]
```
Out[12]= {0}

Claim 2: For any kappa in W we have kappa = σ(slashKappa), where slashKappa is the “first trace” of kappa. That is, the trace-free (1,1)-tensor

L^i_j = kappa^is_js - 1/2 Trace(kappa) delta^i_j.

- **Step 1: Define arbitrary kappa**

In[13]:=
```
kMat = {{k11, k12, k13, k14, k15, k16},
        {k21, k22, k23, k24, k25, k26},
        {k31, k32, k33, k34, k35, k36},
        {k41, k42, k43, k44, k45, k46},
        {k51, k52, k53, k54, k55, k56},
        {k61, k62, k63, k64, k65, k66}};

kappa = emMatrixToKappa[kMat];
```

- **Step 2: Derive conditions that coefficients of kappa should satisfy when kappa is purely skewon (that is in W)**

In[15]:=
```
conditions = emSkewonConditions[kappa]
```

Out[15]=
```
{0, -4 k14, 4 k14, -2 k15 - 2 k24, 2 k15 + 2 k24, -4 k25, 4 k25, -2 k16 - 2 k34, 2 k16 + 2 k34,
 -2 k26 - 2 k35, 2 k26 + 2 k35, -4 k36, 4 k36, -4 k41, 4 k41, -2 k11 - 2 k44, 2 k11 + 2 k44,
 -2 k21 - 2 k45, 2 k21 + 2 k45, -2 k31 - 2 k46, 2 k31 + 2 k46, -2 k42 - 2 k51, 2 k42 + 2 k51,
 -4 k52, 4 k52, -2 k12 - 2 k54, 2 k12 + 2 k54, -2 k22 - 2 k55, 2 k22 + 2 k55, -2 k32 - 2 k56,
 2 k32 + 2 k56, -2 k43 - 2 k61, 2 k43 + 2 k61, -2 k53 - 2 k62, 2 k53 + 2 k62, -4 k63, 4 k63,
 -2 k13 - 2 k64, 2 k13 + 2 k64, -2 k23 - 2 k65, 2 k23 + 2 k65, -2 k33 - 2 k66, 2 k33 + 2 k66}
```

- **Step 3: Compute kappa2 = sigma(slashKappa) and show that kappa = kappa2**

In[16]:=
```
emSlashKappa[kappa_] := Module[{i, j},
   Table[
     Sum[emReadNormal[kappa, i, s, j, s], {s, 1, 4}],
     {i, 1, 4}, {j, 1, 4}
    ] - 2 emTrace[kappa] 1 / 4 IdentityMatrix[4]
  ]
```

In[17]:=
```
Simplify[Tr[emSlashKappa[kappa]]]
```

Out[17]= 0

In[18]:=
```
kappa2 = Simplify[sigma[emSlashKappa[kappa]]];
emKappaToMatrix[kappa2] // MatrixForm
```

Out[19]//MatrixForm=

$$\begin{pmatrix} \frac{k11-k44}{2} & \frac{k12-k54}{2} & \frac{k13-k64}{2} & 0 & \frac{k15-k24}{2} & \frac{k16-k34}{2} \\ \frac{k21-k45}{2} & \frac{k22-k55}{2} & \frac{k23-k65}{2} & \frac{1}{2}(-k15+k24) & 0 & \frac{k26-k35}{2} \\ \frac{k31-k46}{2} & \frac{k32-k56}{2} & \frac{k33-k66}{2} & \frac{1}{2}(-k16+k34) & \frac{1}{2}(-k26+k35) & 0 \\ 0 & \frac{k42-k51}{2} & \frac{k43-k61}{2} & \frac{1}{2}(-k11+k44) & \frac{1}{2}(-k21+k45) & \frac{1}{2}(-k31+k46) \\ \frac{1}{2}(-k42+k51) & 0 & \frac{k53-k62}{2} & \frac{1}{2}(-k12+k54) & \frac{1}{2}(-k22+k55) & \frac{1}{2}(-k32+k56) \\ \frac{1}{2}(-k43+k61) & \frac{1}{2}(-k53+k62) & 0 & \frac{1}{2}(-k13+k64) & \frac{1}{2}(-k23+k65) & \frac{1}{2}(-k33+k66) \end{pmatrix}$$

In[20]:=
```
eqs = Union[Flatten[kappa2 - kappa]]
```

Out[20]=

$$\Big\{0,\ -k14,\ k15+\frac{1}{2}(-k15+k24),\ -k24+\frac{1}{2}(-k15+k24),\ k25,\ -k16+\frac{k16-k34}{2},$$
$$-k34+\frac{1}{2}(-k16+k34),\ -k26+\frac{k26-k35}{2},\ \frac{k26-k35}{2}+k35,\ -k36,\ -k41,\ -k11+\frac{k11-k44}{2},$$
$$-k44+\frac{1}{2}(-k11+k44),\ -k21+\frac{k21-k45}{2},\ \frac{k21-k45}{2}+k45,\ -k31+\frac{k31-k46}{2},$$
$$-k46+\frac{1}{2}(-k31+k46),\ -k42+\frac{k42-k51}{2},\ \frac{k42-k51}{2}+k51,\ k52,\ -k12+\frac{k12-k54}{2},$$
$$\frac{k12-k54}{2}+k54,\ -k22+\frac{k22-k55}{2},\ -k55+\frac{1}{2}(-k22+k55),\ -k32+\frac{k32-k56}{2},$$
$$\frac{k32-k56}{2}+k56,\ -k43+\frac{k43-k61}{2},\ -k61+\frac{1}{2}(-k43+k61),\ k53+\frac{1}{2}(-k53+k62),$$
$$-k62+\frac{1}{2}(-k53+k62),\ -k63,\ -k13+\frac{k13-k64}{2},\ -k64+\frac{1}{2}(-k13+k64),$$
$$-k23+\frac{k23-k65}{2},\ \frac{k23-k65}{2}+k65,\ -k33+\frac{k33-k66}{2},\ -k66+\frac{1}{2}(-k33+k66)\Big\}$$

In[21]:=
```
Union[Simplify[eqs, toEqs[conditions]]]
```

Out[21]= {0}

Claim 3: If σ(eta)=0 then eta = 0.

In[22]:=
```
eta = (w11 w12 w13 w14
       w21 w22 w23 w24
       w31 w32 w33 w34
       w41 w42 w43 w44);

eta0 = eta - 1 / 4 Tr[eta] IdentityMatrix[4];
Tr[eta0]
```

Out[24]= 0

In[25]:=
```
(* We assume that σ(eta0)=0. *)
eqs = Simplify[Union[Flatten[sigma[eta0]]]]
```

Out[25]= $\left\{0, \frac{w12}{2}, -\frac{w13}{2}, \frac{w13}{2}, -\frac{w14}{2}, \frac{w21}{2}, \frac{w23}{2}, -\frac{w24}{2}, \frac{w24}{2}, -\frac{w31}{2}, \frac{w31}{2}, \frac{w32}{2}, \frac{w34}{2}, -\frac{w41}{2}, -\frac{w42}{2}, \frac{w42}{2}, \frac{w43}{2}, \frac{1}{4}(w11 + w22 - w33 - w44), \frac{1}{4}(w11 - w22 + w33 - w44), \frac{1}{4}(-w11 + w22 + w33 - w44), \frac{1}{4}(w11 - w22 - w33 + w44), \frac{1}{4}(-w11 + w22 - w33 + w44), \frac{1}{4}(-w11 - w22 + w33 + w44)\right\}$

In[26]:=
```
Union[Flatten[Simplify[eta0, toEqs[eqs]]]]
```

Out[26]= {0}

- **Claim 4: If kappa is an arbitrary (twisted) (2,2)-tensor and**

  kappa III = 1/6 trace(kappa) Id
  kappa II = sigma(slashKappa(kappa))
  kappaI = kappa - kappaII - kappaIII

  then kappaI is in Z, kappaII is in W and kappaIII is in U.

In[27]:=
```
kMat = (k11 k12 k13 k14 k15 k16
        k21 k22 k23 k24 k25 k26
        k31 k32 k33 k34 k35 k36
        k41 k42 k43 k44 k45 k46
        k51 k52 k53 k54 k55 k56
        k61 k62 k63 k64 k65 k66);

kappa = emMatrixToKappa[kMat];
```

In[29]:=
```
kappaIII = 1 / 6 emTrace[kappa] emIdentityKappa[];
kappaII = sigma[emSlashKappa[kappa]];
kappaI = kappa - kappaII - kappaIII;
```

In[32]:=
```
Union[Simplify[emSkewonConditions[kappaII]]]
Union[Simplify[emPrincipalConditions[kappaI]]]
```

Out[32]= {0}

Out[33]= {0}